

Math 2280-2

Example 5.6.4

```
> with(LinearAlgebra):
```

System matrix

```
> A:= Matrix(2,2,[4,2,3,-1]);
```

$$A := \begin{bmatrix} 4 & 2 \\ 3 & -1 \end{bmatrix} \quad (1)$$

Initial condition

```
> x0 := Vector([7,3]);
```

$$x_0 := \begin{bmatrix} 7 \\ 3 \end{bmatrix} \quad (2)$$

non-homogeneous term

```
> f := t -> -t*exp(-2*t)*Vector([15,4]);
```

$$f := t \mapsto -t e^{-2t} \text{Vector}([15, 4]) \quad (3)$$

Take matrix exponential to get fundamental matrix solution

```
> expAt := t -> MatrixExponential(A*t);
```

$$\text{expAt} := t \mapsto \text{LinearAlgebra-MatrixExponential}(A t) \quad (4)$$

```
> expAt(t);
```

$$\begin{bmatrix} \frac{6}{7} e^{5t} + \frac{1}{7} e^{-2t} & -\frac{2}{7} e^{-2t} + \frac{2}{7} e^{5t} \\ -\frac{3}{7} e^{-2t} + \frac{3}{7} e^{5t} & \frac{1}{7} e^{5t} + \frac{6}{7} e^{-2t} \end{bmatrix} \quad (5)$$

Form integrand in formula (28)

```
> integrand := t->MatrixInverse(expAt(t)) . f(t);
```

$$\text{integrand} := t \mapsto \text{LinearAlgebra-MatrixInverse}(\text{expAt}(t)) \cdot f(t) \quad (6)$$

And the integral in formula 28 would be in maple (map means "apply function to each element of matrix")

```
> map(int,integrand(s),s=0..t);
```

$$\begin{bmatrix} -\frac{1}{14} (4 e^{7t} - 28 t - 4 + 7 t^2 e^{7t}) e^{-7t} \\ \frac{1}{14} (-2 e^{7t} + 21 t^2 e^{7t} + 14 t + 2) e^{-7t} \end{bmatrix} \quad (7)$$

Actually do formula (28)

```
> sol := t-> expAt(t) . ( x0 + map(int,integrand(s),s=0..t));
```

$$\text{sol} := t \mapsto \text{expAt}(t) \cdot (x_0 + \text{map}(\text{int}, \text{integrand}(s), s=0..t)) \quad (8)$$

```
> simplify(sol(t));
```

(9)

$$\begin{bmatrix} \frac{46}{7} e^{5t} + 2 e^{-2t} t + \frac{3}{7} e^{-2t} - \frac{1}{2} e^{-2t} t^2 \\ -\frac{2}{7} e^{-2t} + \frac{3}{2} e^{-2t} t^2 + \frac{23}{7} e^{5t} + e^{-2t} t \end{bmatrix} \quad (9)$$

Reality check: does our solution satisfy DE?

$$\begin{aligned} &> \text{ **simplify(map(diff,sol(t),t) - A . sol(t) - f(t));** } \\ &\quad \begin{bmatrix} 0 \\ 0 \end{bmatrix} \end{aligned} \quad (10)$$

>